TI-80

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists.

```
PROGRAM:SOLVE
:DISP “AX+BY=C”
:INPUT “ENTER A”,A
:INPUT “ENTER B”,B
:INPUT “ENTER C”,C
:DISP “DX+EY=F”
:INPUT “ENTER D”,D
:INPUT “ENTER E”,E
:INPUT “ENTER F”,F
:IF AE–DB=0
:THEN
:DISP “NO UNIQUE”
:DISP “SOLUTION”
:ELSE
:(CE–BF)/(AE–DB)→X
:(AF–CD)/(AE–DB)→Y
:DISP X
:DISP Y
:END
```

Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
PROGRAM:ADDVECT
:CLRDRAW
:DISP “ENTER(A,B)”
:INPUT “ENTER A”,A
:INPUT “ENTER B”,B
:DISP “ENTER (C,D)”
:INPUT “ENTER C”,C
:INPUT “ENTER D”,D
:LINE(0,0,A,B)
:LINE(0,0,C,D)
:A+C→E
:B+D→F
:LINE(0,0,E,F)
:LINE(A,B,E,F)
:LINE(C,D,E,F)
:PAUSE
```

Visualizing Row Operations Program not available

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists.

```
Prgm1:SOLVE
:Disp "AX+BY=C"
:Input A
:Input B
:Input C
:Disp "DX+EY=F"
:Input D
:Input E
:Input F
:If AE–DB=0
:Goto 1
:(CE–BF)/(AE–DB)→X
:(AF–CD)/(AE–DB)→Y
:Disp X
:Disp Y
:End
:Lbl 1
:Disp "NO UNIQUE SOLUTION"
:End
```

Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
Prgm2:ADDVECT
:ClrDraw
:Disp "ENTER(A,B)"
:Disp "ENTER A"
:Input A
:Disp "ENTER B"
:Input B
:Disp "ENTER (C,D)"
:Disp "ENTER C"
:Input C
:Disp "ENTER D"
:Input D
:Line(0,0,A,B)
:Line(0,0,C,D)
:A+C→E
:B+D→F
:Line(0,0,E,F)
:Line(A,B,E,F)
:Line(C,D,E,F)
:Pause
:End
```

Visualizing Row Operations Program

This program demonstrates how elementary matrix row operations used in Gauss-Jordan elimination may be interpreted graphically. It asks the user to enter a 2×3 matrix that corresponds to a system of two linear equations. (The matrix entries should not be equivalent to either vertical or horizontal lines. This demonstration is also most effective if the y-intercepts of the lines are between -10 and 10.)

While the demonstration is running, you should notice that each elementary row operation creates an equivalent system. This equivalence is reinforced graphically because, although the equations of the lines change with each elementary row operation, the point of intersection remains the same. You may want to run this program a second time to notice the relationship between the row operations and the graphs of the lines of the system. To use this program, dimension matrix [A] as a 2×3 matrix. Press ENTER after each screen display to continue the program.

```
Prgm3:ROWOPS
:Disp "ENTER A"
:Disp "2 BY 3 MATRIX"
:Disp "A B C"
:Disp "D E F"
:Input A
:Input B
:Input C
:Input D
:Input E
:Input F
:A→[A](1,1)
:B→[A](1,2)
:C→[A](1,3)
:D→[A](2,1)
:E→[A](2,2)
:F→[A](2,3)
:ClrHome
:Disp "ORIGINAL MATRIX"
:Disp [A]
:Pause
:"B^-1(C−AX)"→Y2
:"E^-1(F−DX)"→Y1
:-10→Xmin
:10→Xmax
:1→Xscl
:-10→Ymin
:10→Ymax
:1→Yscl
:DispGraph
:Pause
:ClrHome
:Disp "OBTAIN LEADING"
:Disp "1 IN ROW 1"
:*row(A^-1,[A],1)→[A]
:Disp [A]
:Pause
:ClrDraw
:"(A/B)(C/A−X)"→Y2
:DispGraph
:Pause
:ClrHome
:Disp "OBTAIN 0 BELOW"
:Disp "LEADING 1 IN"
:Disp "COLUMN 1"
:*row+(-D,[A],1,2)→[A]
:Disp[A]
:Pause
:ClrDraw
:"(E−(BD/A))^-1(F−(DC/A))"→Y1
:DispGraph
:Pause
:ClrHome
:[A](2,2)→G
:If G=0
:Goto 1
:*row(G^-1,[A],2)→[A]
:Disp "OBTAIN LEADING"
:Disp "1 IN ROW 2"
:Disp [A]
:Pause
:ClrDraw
:DispGraph
:Pause
:ClrHome
:Disp "OBTAIN 0 ABOVE"
:Disp "LEADING 1 IN"
:Disp "COLUMN 2"
:[A](1,2)→H
:*row+(-H,[A],2,1)→[A]
:Disp [A]
:Pause
:ClrDraw
:Y2−Off
:Line([A](1,3),-10,[A](1,3),10)
:DispGraph
:Pause
:ClrHome
:Disp "THE POINT OF"
:Disp "INTERSECTION IS"
:Disp "X="
:Disp [A](1,3)
:Disp "Y="
:Disp [A](2,3)
:End
:Lbl 1
:If [A](2,3)=0
:Disp "INFINITELY MANY"
:Disp "SOLUTIONS"
:If [A](2,3) ≠ 0
:Disp "INCONSISTENT"
:Disp "SYSTEM"
:End
```

TI-82
TI-83

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists.

```
PROGRAM:SOLVE
:Disp "AX+BY=C"
:Prompt A
:Prompt B
:Prompt C
:Disp "DX+EY=F"
:Prompt D
:Prompt E
:Prompt F
:If AE–DB=0
:Then
:Disp "NO UNIQUE"
:Disp "SOLUTION"
:Else
:(CE–BF)/(AE–DB)→X
:(AF–CD)/(AE–DB)→Y
:Disp X
:Disp Y
:End
```

Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
PROGRAM:ADDVECT
:ClrDraw
:Input "ENTER A",A
:Input "ENTER B",B
:Input "ENTER C",C
:Input "ENTER D",D
:Line(0,0,A,B)
:Line(0,0,C,D)
:A+C→E
:B+D→F
:Line(0,0,E,F)
:Line(A,B,E,F)
:Line(C,D,E,F)
:Pause
:Stop
```

Visualizing Row Operations Program

This program demonstrates how elementary matrix row operations used in Gauss-Jordan elimination may be interpreted graphically. It asks the user to enter a 2×3 matrix that corresponds to a system of two linear equations. (The matrix entries should not be equivalent to either vertical or horizontal lines. This demonstration is also most effective if the *y*-intercepts of the lines are between -10 and 10.)

While the demonstration is running, you should notice that each elementary row operation creates an equivalent system. This equivalence is reinforced graphically because, although the equations of the lines change with each elementary row operation, the point of intersection remains the same. You may want to run this program a second time to notice the relationship between the row operations and the graphs of the lines of the system. To use this program, dimension matrix [A] as a 2×3 matrix. Press ENTER after each screen display to continue the program.

```
PROGRAM: ROWOPS
:Disp "ENTER A"
:Disp "2 BY 3 MATRIX:"
:Disp "A B C"
:Disp "D E F"
:Prompt A,B,C
:Prompt D,E,F
:A→[A](1,1):B→[A](1,2)
:C→[A](1,3):D→[A](2,1)
:E→[A](2,2):F→[A](2,3)
:ClrHome
:Disp "ORIGINAL MATRIX:"
:Pause [A]
:"B⁻¹(C−AX)"→Y2
:"E⁻¹(F−DX)"→Y1
:ZStandard:Pause:ClrHome
:Disp "OBTAIN LEADING"
:Disp "1 IN ROW 1"
:*row(A⁻¹,[A],1)→[A]
:Pause [A]:ClrDraw
:"(A/B)(C/A−X)"→Y2
:DispGraph:Pause:ClrHome
:Disp "OBTAIN 0 BELOW"
:Disp "LEADING 1 IN"
:Disp "COLUMN 1"
:*row+(-D,[A],1,2)→[A]
:Pause [A]:ClrDraw
:"(E−(BD/A))⁻¹(F−(DC/A))"→Y1
:DispGraph:Pause:ClrHome
:[A](2,2)→G
:If G=0
:Goto 1
:*row(G⁻¹,[A],2)→[A]
:Disp "OBTAIN LEADING"
:Disp "1 IN ROW 2"
:Pause [A]:ClrDraw
:DispGraph:Pause:ClrHome
:Disp "OBTAIN 0 ABOVE"
:Disp "LEADING 1 IN"
:Disp "COLUMN 2"
:[A](1,2)→H
:*row+(⁻H,[A],2,1)→[A]
:Pause [A]:ClrDraw:FnOff 2
:Vertical ⁻(B/A)(E−(BD/A))⁻¹(F−DC/A)+C/A
:DispGraph:Pause:ClrHome
:Disp "THE POINT OF"
:Disp "INTERSECTION IS"
:Disp "X=",[A](1,3),"Y=",[A](2,3)
:Stop
:Lbl 1
:If [A](2,3)=0
:Then
:Disp "INFINITELY MANY"
:Disp "SOLUTIONS"
:Else
:Disp "INCONSISTENT"
:Disp "SYSTEM"
:End
```

TI-85
TI-86

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists.

```
PROGRAM:SOLVE
:Disp "AX+BY=C"
:Input "ENTER A",A
:Input "ENTER B",B
:Input "ENTER C",C
:Disp "DX+EY=F"
:Input "ENTER D",D
:Input "ENTER E",E
:Input "ENTER F",F
:If A*E–D*B==0
:Goto A
:(C*E–B*F)/(A*E–D*B)→X
:(A*F–C*D)/(A*E–D*B)→Y
:Disp X
:Disp Y
:Stop
:Lbl A
:Disp "NO UNIQUE SOLUTION"
```

Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
PROGRAM:ADDVECT
:ClDrw
:Input "enter A",A
:Input "enter B",B
:Input "enter C",C
:Input "enter D",D
:Line(0,0,A,B)
:Line(0,0,C,D)
:A+C→E
:B+D→F
:Line(0,0,E,F)
:Line(A,B,E,F)
:Line(C,D,E,F)
:Pause
:Stop
```

Visualizing Row Operations Program

This program demonstrates how elementary matrix row operations used in Gauss-Jordan elimination may be interpreted graphically. It asks the user to enter a 2×3 matrix that corresponds to a system of two linear equations. (The matrix entries should not be equivalent to either vertical or horizontal lines. This demonstration is also most effective if the y-intercepts of the lines are between -10 and 10.)

While the demonstration is running, you should notice that each elementary row operation creates an equivalent system. This equivalence is reinforced graphically because, although the equations of the lines change with each elementary row operation, the point of intersection remains the same. You may want to run this program a second time to notice the relationship between the row operations and the graphs of the lines of the system. To use this program, dimension matrix TEMP as a 2×3 matrix. Press ENTER after each screen display to continue the program.

```
PROGRAM: ROWOPS
:Disp “ENTER A”
:Disp “2 BY 3 MATRIX:”
:Disp “A B C”
:Disp “D E F”
:Prompt A,B,C
:Prompt D,E,F
:A→TEMP(1,1):B→TEMP(1,2)
:C→TEMP(1,3):D→TEMP(2,1)
:E→TEMP(2,2):F→TEMP(2,3)
:ClLCD
:Disp “ORIGINAL MATRIX:”
:Disp TEMP
:Pause
:y2=B⁻¹(C−A∗x)
:y1=E⁻¹(F−D∗x)
:ZStd:Pause:ClLCD
:Disp “OBTAIN LEADING”
:Disp “1 IN ROW 1”
:multR(A⁻¹,TEMP,1)→TEMP
:Disp TEMP:Pause
:“(A/B)(C/A−x)”→y
:ClDrw:DispG:Pause:ClLCD
:Disp “OBTAIN 0 BELOW”
:Disp “LEADING 1 IN”
:Disp “COLUMN 1”
:mRAdd(-D,TEMP,1,2)→TEMP
:Disp TEMP:Pause:
:If TEMP(2,2)==0
:Goto A
:y1=(E−(B∗D/A))⁻¹(F−(D∗C/A))
:ClDrw:DispG:Pause:ClLCD
:TEMP(2,2)→G
:multR(G⁻¹,TEMP,2)→TEMP
:Disp “OBTAIN LEADING”
:Disp “1 IN ROW 2”
:Disp TEMP:Pause:ClDrw
:DispG:Pause:ClLCD
:Disp “OBTAIN 0 ABOVE”
:Disp “LEADING 1 IN”
:Disp “COLUMN 2”
:TEMP(1,2)→H
:mRAdd(-H,TEMP,2,1)→TEMP
:Disp TEMP:Pause:FnOff 2:ClDrw
:Vert -(B/A)(E−(B∗D/A))⁻¹(F−D∗C/A)+C/A
:DispG:Pause:ClLCD
:Disp “THE POINT OF”
:Disp “INTERSECTION IS”
:Disp “X=”,TEMP(1,3),“Y=”,TEMP(2,3)
:Stop
:Lbl A
:If TEMP(2,3)==0
:Then
:Disp “INFINITELY MANY”
:Disp “SOLUTIONS”
:Else
:Disp “INCONSISTENT”
:Disp “SYSTEM”
:End
```

TI-92

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists.

```
:solvelin( )
:Prgm
:ClrI0
:Disp "Ax+By=C"
:Input "Enter A.",a
:Input "Enter B.",b
:Input "Enter C.",c
:ClrI0
:Disp "Dx+Ey=F"
:Input "Enter D.",d
:Input "Enter E.",e
:Input "Enter F.",f
:If a∗e–d∗b=0 Then
:       Disp "No unique solution"
:   Else
:       (c∗e–b∗f)/(a∗e–d∗b)→x
:       (a∗f–c∗d)/(a∗e–d∗b)→y
:       Disp x
:       Disp y
:EndIf
:EndPrgm
```

Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
:addvect( )
:Prgm
:ClrIO
:Input "ENTER a",a
:Input "ENTER b",b
:Input "ENTER c",c
:Input "ENTER d",d
:ClrDraw
:Line(0,0,a,b)
:Line(0,0,c,d)
:a+c→e
:b+d→f
:Line 0,0,e,f
:Line a,b,e,f
:Line c,d,e,f
:Pause
:setMode("Split 1 App","Home")
:Stop
:EndPrgm
```

Visualizing Row Operations Program

This program demonstrates how elementary matrix row operations used in Gauss-Jordan elimination may be interpreted graphically. It asks the user to enter a 2×3 matrix that corresponds to a system of two linear equations. (The matrix entries should not be equivalent to either vertical or horizontal lines. This demonstration is also most effective if the y-intercepts of the lines are between -10 and 10.)

While the demonstration is running, you should notice that each elementary row operation creates an equivalent system. This equivalence is reinforced graphically because, although the equations of the lines change with each elementary row operation, the point of intersection remains the same. You may want to run this program a second time to notice the relationship between the row operations and the graphs of the lines of the system. Press ENTER after each screen display to continue the program.

```
:rowops( )
:Prgm
:ClrIO
:ClrHome
:setMode("Split Screen","Left-Right")
:setMode("Split 1 App","Home")
:setMode("Split 2 App","Graph")
:Disp "ENTER A"
:Disp "2 BY 3 MATRIX:"
:Disp "A B C"
:Disp "D E F"
:Prompt a,b,c
:Prompt d,e,f
:[[a,b,c][d,e,f]]→mat1
:ClrIO
:b^(-1)∗(c-a∗x)→y2(x)
:e^(-1)∗(f-d∗x)→y1(x)
:ZoomStd
:Disp "ORIGINAL MATRIX:"
:Pause mat1
:ClrIO
:a/b∗(c/a-x)→y2(x)
:Disp "OBTAIN LEADING"
:Disp "1 IN ROW 1"
:mRow(a^(-1),mat1,1)→mat1
:Pause mat1
:ClrIO
:(e-b∗d/a)^(-1)∗(f-d∗c/a)→y1(x)
:DispG
:Disp "OBTAIN 0 BELOW"
:Disp "LEADING 1 IN"
:Disp "COLUMN 1"
:mRowAdd(-d,mat1,1,2)→mat1
:Pause mat1
:ClrIO
:mat1[2,2]→g
:If g=0
:Goto a1
:Disp "OBTAIN LEADING"
:Disp "1 IN ROW 2"
:mRow(g^(-1),mat1,2)→mat1
:Pause mat1
:ClrIO
:mat1[1,2]→h
:FnOff 2
:LineVert -b/a∗(e-b∗d/a)^(-1)∗(f-d∗c/a)+c/a
:Disp "OBTAIN 0 ABOVE"
:Disp "LEADING 1 IN"
:Disp "COLUMN 2"
:mRowAdd(-h,mat1,2,1)→mat1
:Pause mat1
:ClrIO
:Disp "THE POINT OF"
:Disp "INTERSECTION IS"
:Disp "X=",mat1[1,3],"Y=",mat1[2,3]
:Goto A2
:Lbl a1
:If mat1[2,3]=0 Then
:Disp "INFINITELY MANY"
:Disp "SOLUTIONS"
:Else
:Disp "INCONSISTENT"
:Disp "SYSTEM"
:EndIf
:Lbl A2
:Pause
:setMode("Split Screen","Full")
:EndPrgm
```

Casio fx-7700G

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists.

```
SOLVE
"AX+BY=C"
"A="?→A
"B="?→B
"C="?→C
"DX+EY=F"
"D="?→D
"E="?→E
"F="?→F
AE–DB=0⇒Goto 1
"X=":(CE−BF)÷(AE−DB)◢
"Y=":(AF−CD)÷(AE−DB)
Goto 2
Lbl 1
"NO UNIQUE SOLUTION"
Lbl 2
```

Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
ADDVECT
Cls
"A="?→A
"B="?→B
"C="?→C
"D="?→D
Plot 0,0
Plot A,B
Line
Plot 0,0
Plot C,D
Line◢
A+C→E
B+D→F
Plot 0,0
Plot E,F
Line
Plot A,B
Plot E,F
Line
Plot C,D
Plot E,F
Line◢
```

Visualizing Row Operations Program not available

Casio fx-7700GE
Casio fx-9700GE
Casio CFX-9800G
Casio CFX-9850G

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists. Solutions to systems of linear equations are also available directly from the Casio calculator's EQUATION MENU.

```
SOLVE
"AX+BY=C"↵
"A=":?→A↵
"B=":?→B↵
"C=":?→C↵
"DX+EY=F"↵
"D=":?→D↵
"E=":?→E↵
"F=":?→F↵
AE–DB=0⇒Goto 1↵
"X=":(CE–BF)÷(AE–DB)◢
"Y=":(AF–CD)÷(AE–DB)↵
Goto 2↵
Lbl 1↵
"NO UNIQUE SOLUTION"↵
Lbl 2
```

Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
ADDVECT
Cls↵
"A="?→A↵
"B="?→B↵
"C="?→C↵
"D="?→D↵
Plot 0,0↵
Plot A,B↵
Line↵
Plot 0,0↵
Plot C,D↵
Line↵
A+C→E↵
B+D→F↵
Plot 0,0↵
Plot E,F↵
Line↵
Plot A,B↵
Plot E,F↵
Line↵
Plot C,D↵
Plot E,F↵
Line◢
```

Visualizing Row Operations Program

This program demonstrates how elementary matrix row operations used in Gauss-Jordan elimination may be interpreted graphically. It asks the user to enter a 2×3 matrix that corresponds to a system of two linear equations. (The matrix entries should not be equivalent to either vertical or horizontal lines. This demonstration is also most effective if the y-intercepts of the lines are between -10 and 10.)

While the demonstration is running, you should notice that each elementary row operation creates an equivalent system. This equivalence is reinforced graphically because, although the equations of the lines change with each elementary row operation, the point of intersection remains the same. You may want to run this program a second time to notice the relationship between the row operations and the graphs of the lines of the system. To use this program, dimension Mat A as a 2×3 matrix. Press EXE after each screen display to continue the program.

```
ROWOPS
"ENTER A"↵
"2 BY 3 MATRIX:"↵
"A B C"↵
"D E F"↵
"A="?→A:"B="?→B:
"C="?→C:"D="?→D:
"E="?→E:"F="?→F:↵
[[A,B,C][D,E,F]]→Mat A↵
Cls↵
"ORIGINAL MATRIX:"◢
Mat A◢
Range -10,10,1,-10,10,1↵
Graph Y=B⁻¹(C−AX)↵
Graph Y=E⁻¹(F−DX)◢
Cls↵
"OBTAIN LEADING"↵
"1 IN ROW 1"◢
*Row A⁻¹,A,1↵
Mat A◢
Graph Y=(A÷B)(C÷A−X)↵
Graph Y=E⁻¹(F−DX)◢
Cls↵
"OBTAIN 0 BELOW"↵
"LEADING 1 IN"↵
"COLUMN 1"◢
*Row+ -D,A,1,2↵
Mat A◢
Graph Y=(A÷B)(C÷A−X)↵
GraphY=(E−(BD÷A))⁻¹(F−(DC÷A))◢
Cls↵
Mat A[2,2]→G↵
G=0⇒Goto 1↵
*Row G⁻¹,A,2↵
"OBTAIN LEADING"↵
"1 IN ROW 2"◢
Mat A◢
Graph Y=(A÷B)(C÷A−X)↵
Graph Y=(E−(BD÷A))⁻¹(F−(DC÷A))◢
Cls↵
"OBTAIN 0 ABOVE"↵
"LEADING 1 IN"↵
"COLUMN 2"◢
Mat A[1,2]→H↵
*Row+ -H,A,2,1↵
Mat A◢
Mat A[1,3]→J↵
Mat A[2,3]→K↵
Graph Y=K↵
Plot J,-10:Plot J,10:Line◢
"THE POINT OF"↵
"INTERSECTION IS"↵
"X=":J◢
"Y=":K
Goto 3↵
Lbl 1↵
Mat A[2,3]=0⇒Goto 2↵
"INCONSISTENT"↵
"SYSTEM"↵
Goto 3↵
Lbl 2↵
"INFINITELY MANY"↵
"SOLUTIONS"↵
Lbl 3
```

Sharp EL-9200C
Sharp EL-9300C

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists. Equations must be entered in the form: AX + BY = C; DX + EY = F. Uppercase letters are used so that the values can be accessed in the calculation mode of the calculator.

```
solve
————REAL
Print "AX+BY=C"
Input A
Input B
Input C
Print "DX+EY=F"
Input D
Input E
Input F
If A*E−D*B=0 Goto 1
X=(C*E−B*F)/(A*E−D*B)
Y=(A*F−C*D)/(A*E−D*B)
Print X
Print Y
End
Label 1
Print "no unique solution"
End
```

Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
addvect
————REAL
ClrG
Input a
Input b
Input c
Input d
Line 0,0,a,b
Line 0,0,c,d
e=a+c
f=b+d
Line 0,0,e,f
Line a,b,e,f
Line c,d,e,f
Wait
End
```

Visualizing Row Operations Program not available

HP-38G

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists.

1. Input the 2 programs SOLVE and SOLVE.SOLN.
2. Run the SOLVE program.

```
SOLVE
SOLVE PROGRAM
INPUT A;“AX+BY=C”;
   “ENTER A”;“ ”;1:
INPUT B;“AX+BY=C”;
   “ENTER B”;“ ”;1:
INPUT C;“AX+BY=C”;
   “ENTER C”;“ ”;1:
INPUT D;“DX+EY=F”;
   “ENTER D”;“ ”;1:
INPUT E;“DX+EY=F”;
   “ENTER E”;“ ”;1:
INPUT F;“DX+EY=F”;
   “ENTER F”;“ ”;1:
ERASE:
IF AE−DB==0
THEN DISP 3; “NO UNIQUE  SOLUTION”:
ELSE RUN “SOLVE.SOLN”:
END:
FREEZE:
SOLVE.SOLN PROGRAM
(CE−BF)/(AE−DB)►X:
(AF−CD)/(AE−DB)►Y:
DISP 3;“X=”X:
DISP 5;“Y=”Y:
```

Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle. The Function aplet should have a plot range of $-10 \le x \le 10$ and $-10 \le y \le 10$.

```
ADDVECT PROGRAM
INPUT A;; “ENTER A”;;1:
INPUT B;; “ENTER B”;;1:
INPUT C;; “ENTER C”;;1:
INPUT D;; “ENTER D”;;1:
ERASE:
LINE−10;0;10;0:
LINE 0;−10;0;10:
LINE 0;0;A;B:
LINE 0;0;C;D:
FREEZE:
A+C► E
B+D► F
LINE 0;0;E;F:
LINE A;B;E;F:
LINE C;D;E;F:
FREEZE
```

Visualizing Row Operations Program not available